



*presents the*

## REFERENCE IMPLEMENTATION

also known as the  
**PURPLE PAPER**

*of the remarkable*

## DAI SYSTEM

a decentralized stability mechanism

*formulated by*

Daniel Brockman  
Mikael Brockman  
Nikolai Mushegian

*with last update on April 17, 2017.*



# Contents

|          |   |           |
|----------|---|-----------|
| <b>1</b> | <b>Introduction</b>                                 | <b>5</b>  |
| 1.1      | Naming . . . . .                                    | 5         |
| 1.2      | Motivation . . . . .                                | 6         |
| 1.3      | Limitations . . . . .                               | 7         |
| 1.4      | Verification . . . . .                              | 7         |
| <br>     |   |           |
| <b>I</b> | <b>Implementation</b>                               | <b>8</b>  |
| <br>     |   |           |
| <b>2</b> | <b>Preamble</b>                                     | <b>9</b>  |
| <br>     |   |           |
| <b>3</b> | <b>Types</b>  | <b>10</b> |
| 3.1      | Numeric types . . . . .                             | 10        |
| 3.2      | Identifiers and addresses . . . . .                 | 11        |
| 3.3      | Tag — external token price data . . . . .           | 11        |
|          | tag — market price of token . . . . .               | 11        |
|          | zzz — expiration time of token price feed . . . . . | 11        |
| 3.4      | Entity — token balance holder . . . . .             | 12        |
| 3.5      | Ilk — urn type . . . . .                            | 12        |
|          | gem — assetcoin identifier . . . . .                | 12        |
|          | mat — liquidation ratio . . . . .                   | 12        |
|          | axe — liquidation penalty . . . . .                 | 12        |
|          | hat — issuance ceiling . . . . .                    | 12        |
|          | tax — stability fee factor . . . . .                | 12        |
|          | lax — price feed limbo duration . . . . .           | 12        |
|          | rho — timestamp of fee unit adjustment . . . . .    | 12        |
|          | rum — total outstanding art in fee unit . . . . .   | 12        |
|          | chi — value of fee unit in stablecoin . . . . .     | 12        |
| 3.6      | Urn — stablecoin issuance account . . . . .         | 13        |
|          | cat — address of riddance initiator . . . . .       | 13        |

|          |  |           |
|----------|--|-----------|
|          | lad — urn owner . . . . .                                    | 13        |
|          | ilk — urn type . . . . .                                     | 13        |
|          | art — stablecoin issuance in fee unit . . . . .              | 13        |
|          | ink — amount of locked assetcoin . . . . .                   | 13        |
| 3.7      | Vox — feedback mechanism data . . . . .                      | 13        |
|          | wut — SDR market price of stablecoin . . . . .               | 13        |
|          | par — SDR target price of stablecoin . . . . .               | 13        |
|          | how — sensitivity parameter . . . . .                        | 13        |
|          | way — rate of target price change . . . . .                  | 13        |
|          | tau — time of latest feedback cycle . . . . .                | 13        |
| 3.8      | Vat — system root . . . . .                                  | 13        |
| 3.9      | System model . . . . .                                       | 14        |
|          | era — current time . . . . .                                 | 14        |
| 3.10     | Default data . . . . .                                       | 14        |
| <b>4</b> | <b>Acts</b> . . . . .  | <b>16</b> |
| 4.1      | Assessment . . . . .   | 17        |
|          | feel — identify urn risk stage . . . . .                     | 17        |
| 4.2      | Issuance . . . . .   | 19        |
|          | open — create urn . . . . .                                  | 19        |
|          | give — change urn owner . . . . .                            | 19        |
|          | lock — lock assetcoin . . . . .                              | 19        |
|          | free — reclaim assetcoin . . . . .                           | 20        |
|          | draw — issue stablecoin . . . . .                            | 20        |
|          | wipe — reverse stablecoin issuance . . . . .                 | 21        |
|          | shut — wipe, free, and delete urn . . . . .                  | 21        |
| 4.3      | Adjustment . . . . .   | 23        |
|          | prod — adjust target price and target rate . . . . .         | 23        |
|          | drip — update fee unit and unprocessed fee revenue . . . . . | 24        |
| 4.4      | Price feed input . . . . .                                   | 25        |
|          | mark — update market price of assetcoin . . . . .            | 25        |
|          | tell — update market price of stablecoin . . . . .           | 25        |
| 4.5      | Riddance . . . . .   | 25        |
|          | bite — mark for riddance . . . . .                           | 25        |
|          | grab — take stablecoin and anticon for riddance . . . . .    | 26        |
|          | plop — finish riddance returning profit . . . . .            | 26        |
|          | loot — take unprocessed stability fees . . . . .             | 27        |
| 4.6      | Auctioning . . . . .   | 27        |
|          | flip — put assetcoin up for riddance auction . . . . .       | 27        |
|          | flap — put fee revenue stablecoin up for auction . . . . .   | 27        |
|          | flop — inflate countercoin for auction . . . . .             | 27        |

|     |  |    |
|-----|--|----|
| 4.7 | Settlement . . . . .   | 28 |
|     | tidy — burn equal quantities of stablecoin and anticon . . . . . | 28 |
|     | kick — flap, flop, and whatnot . . . . .                         | 28 |
| 4.8 | Governance . . . . .   | 28 |
|     | form — create a new ilk . . . . .                                | 28 |
|     | frob — set the sensitivity parameter . . . . .                   | 29 |
|     | chop — set riddance penalty . . . . .                            | 29 |
|     | cork — set ilk ceiling . . . . .                                 | 29 |
|     | calm — set limbo duration . . . . .                              | 29 |
|     | cuff — set riddance ratio . . . . .                              | 29 |
|     | crop — set stability fee . . . . .                               | 29 |
| 4.9 | Token manipulation . . . . .                                     | 29 |
|     | mint — inflate token . . . . .                                   | 30 |
|     | burn — deflate token . . . . .                                   | 30 |
|     | lend — mint stablecoin and anticon . . . . .                     | 30 |
|     | mend — burn dai and debt token . . . . .                         | 30 |
| 5   | Act framework . . . . .  | 31 |
| 5.1 | The Maker monad . . . . .  | 31 |
| 5.2 | Asserting . . . . .  | 32 |
| A   | Prelude . . . . .  | 33 |
| B   | Fixed point numbers with rounding . . . . .                      | 36 |

# Chapter 1

## Introduction

The DAI SYSTEM, henceforth also “Maker,” is a network of Ethereum contracts defining a token that is subject to a decentralized price stability mechanism. The token is named DAI.

For an overview of the economics of the system, as well as descriptions of governance, off-chain mechanisms, and so on, see the whitepaper.

This document is an executable technical specification of the Maker smart contracts. It is a draft; be aware that the contents will certainly change before launch.

### 1.1 Naming

The implementation is formulated in terms of a parallel vocabulary whose concise words can seem meaningless at first glance (e.g., Urn, par, ink). These words are in fact carefully selected for metaphoric resonance and evocative qualities. Definitions of the words along with mnemonic reminders can be found in the glossary.

We have found that though it requires some initial indoctrination, the Maker jargon is good for development and helps when thinking and talking about the structure and mechanics of the system. Here are some of the reasons:

- The parallel jargon lets us sidestep terminological debates; for example, whether to say “rate of target price change” or “target rate.”
- With decoupled financial and technical vocabularies, we can more flexibly improve one without affecting the other.

- The ability to discuss the system formally, with the financial interpretation partly suspended, has suggested insights that would have been harder to think of inside the normal language.
- The precise and distinctive language makes the structure and logic of the implementation more apparent and easier to formalize.

Some readers may perceive the Maker terminology as unnecessarily obscure despite our apologetics. In that case, we recommend a contrasting look at the Ethereum “yellow paper,” after which this document should appear highly legible.

## 1.2 Motivation

The version of this system that will be deployed on the blockchain is written in Solidity, which is a workable smart contract implementation language. This reference implementation is a model of the behavior of those contracts, written as a “literate” Haskell program. The motivations for such a reference implementation include:

1. **Comparison.** Checking two free-standing implementations against each other is a well-known way of ensuring that they both behave as intended.
2. **Testing.** Haskell lets us use powerful testing tools such as QuickCheck and SmallCheck for comprehensively verifying key properties as a middle ground between unit testing and formal verification.
3. **Explicitness.** Coding the contract behavior in Haskell, a purely functional language, enforces explicit description of aspects which Solidity leaves implicit. For example, a Solidity program can read previously unwritten storage and get back a zero value, whereas in Haskell we must give explicit defaults. The state rollback behavior of failed actions is also explicit in the type of the execution function, which may return an error.
4. **Typing.** While Solidity does have a static type system, it is not expressive enough to encode the distinctions made by our system. In particular, the two different decimal fixed point number types that we use are typed in Solidity with one and the same `uint128` type. In Haskell we can make this distinction explicit.
5. **Formality.** The work of translating a Solidity program into a purely functional program opens up opportunities for certain types of formal verification. In particular, this document will be useful for modelling aspects of the system in a proof assistant like Agda, Idris, Coq, or Isabelle. We can also use logical tools for

Haskell, such as Liquid Haskell (which provides compile time logical property checking) and sbv (a toolkit for model checking and symbolic execution).

6. **Clarity.** An implementation not intended to be deployed on the blockchain is free from concerns about optimizing for gas cost and other factors that make the Solidity implementation less ideal as an understandable specification.
7. **Simulation.** Solidity is highly specific to the Ethereum blockchain environment and as such does not have facilities for interfacing with files or other computer programs. This makes the Solidity implementation of the system less useful for doing simulations of the system's economic, game-theoretic, or statistical aspects.

## 1.3 Limitations

This model is limited in that it has

1. a simplified version of authorization for governance;
2. a simplified version of ERC20 token semantics;
3. no implementation of the decentralized auction contracts; and
4. no 256-bit word limits.

These limitations will be addressed in future revisions.

## 1.4 Verification

Separately from this document, we are developing automatic test suites that generate many, large, and diverse action sequences for property verification. One such property is that the reference implementation exactly matches the on-chain implementation; this is verified through the generation of Solidity test cases with assertions covering the entire state. Other key properties include

- that the target price changes only according to the target rate;
- that the total dai supply is fully accounted for;
- that acts are restricted with respect to risk stage;

along with similar invariants and conditions. A future revision of this document will include formal statements of these properties.

# **Part I**

## **Implementation**



# Chapter 2

## Preamble

This is a Haskell program, and as such makes reference to a background of symbols defined in libraries, as a mathematical paper depends on preestablished theories.

Context should allow the reader to understand most symbols without further reading, but Appendix [A](#) lists and briefly explains each imported type and function.

We replace the default prelude module with our own.

```
module Maker where  
import Prelude ()      Import nothing from Prelude  
import Maker.Prelude   Import everything from Maker Prelude
```

We also import our definition of decimal fixed point numbers, listed in Appendix [B](#).

```
import Maker.Decimal
```

Now we proceed to define the specifics of the Maker system.

# Chapter 3

## Types

This chapter defines the data types used by Maker: numeric types, identifiers, on-chain records, and test model data.

Haskell syntax note: **newtype** defines a type synonym with distinct type identity; **data** creates a record type; and **deriving** creates automatic instances of common functionality.

### 3.1 Numeric types

The system uses two different precisions of decimal fixed point numbers, which we call *wads* and *rays*, having respectively 18 digits of precision (used for token quantities) and 36 digits (used for precise rates and ratios). See Appendix B for details on decimal fixed point numbers and rounding.

```
Define the distinct type of currency quantities
newtype Wad = Wad (Decimal E18)
deriving (Ord, Eq, Num, Real, Fractional, RealFrac)
```

```
Define the distinct type of rates and ratios
newtype Ray = Ray (Decimal E36)
deriving (Ord, Eq, Num, Real, Fractional, RealFrac)
```

We also define a type for time durations in whole seconds, as this is the maximum precision allowed by the Ethereum virtual machine.

```
newtype Sec = Sec Int
deriving (Eq, Ord, Enum, Num, Real, Integral)
```

Haskell number types are not automatically converted, so we convert explicitly with a *cast* function.

```
Convert via fractional  $n/m$  form.  
cast :: (Real a, Fractional b)  $\Rightarrow a \rightarrow b$   
cast = fromRational . toRational
```

## 3.2 Identifiers and addresses

There are several kinds of identifiers used in the system, and we use types to distinguish them. The type parameter *a* creates distinct types; e.g., *Id Foo* and *Id Bar* are incompatible.

```
newtype Id a = Id String  
  deriving (Eq, Ord, Show)
```

We define another type for representing Ethereum account addresses.

```
newtype Address = Address String  
  deriving (Eq, Ord, Show)
```

We also define the different kinds of tokens used by the system.

```
data Gem = Gem String Some assetcoin  
  | DAI      Stablecoin  
  | SIN      Anticoin  
  | MKR      Countercoin  
  deriving (Eq, Ord, Show)
```

## 3.3 Tag — external token price data

The data received from price feeds is categorized by token and stored in *Tag* records.

```
data Tag = Tag {  
  · tag :: Wad,  Market price denominated in SDR  
  · zzz :: Sec   Time of price expiration  
} deriving (Eq, Show)
```

### 3.4 Entity — token balance holder

We use a data type to explicitly distinguish the different entities that can hold a token balance or invoke acts.

```
data Entity = Account Address External holder
  | Jar Token vault
  | Joy Spawning account for stablecoin
  | Woe Spawning account for anticon
  | Ice Holding account for anticon
  | Vow Settler
  | Flipper Assetcoin auctioneer
  | Flapper Stablecoin auctioneer
  | Flopper Countercoin auctioneer
  | Toy Test driver
  | God Omnipotent actor
deriving (Eq, Ord, Show)
```

### 3.5 Ilk — urn type

Each urn belongs to an urn type, specified by an Ilk record. Five parameters, *mat*, *axe*, *hat*, *tax* and *lax*, are set by governance and are known as the *risk parameters*. The rest of the values are used by the system to keep track of the current state. The meaning of each ilk parameter is defined by its interactions in the act definitions of Chapter 4; see the whitepaper for an overview.

```
data Ilk = Ilk {
  · gem :: Gem, Assetcoin identifier
  · lax :: Sec, Grace period after price feed becomes unavailable
  · mat :: Ray, Urn riddance ratio of locked value to issued value
  · axe :: Ray, Urn riddance penalty as fraction of urn issuance
  · hat :: Wad, Maximum total issuance for ilk ("issuance ceiling")
  · tax :: Ray, Stability fee as per-second fraction of urn issuance
  · chi :: Ray, Value of fee unit in stablecoin
  · rho :: Sec, Time of latest fee unit adjustment
  · rum :: Wad Total ilk issuance denominated in fee unit
} deriving (Eq, Show)
```

### 3.6 Urn — stablecoin issuance account

An urn record defines a basic entity through which users interact with the system to issue stablecoin. Each urn belongs to an ilk. The urn records the value of locked assetcoin along with the amount of stablecoin created for this particular urn. When riddance is triggered on an urn, the identity of the triggering entity is also recorded.

```
data Urn = Urn {  
  · ilk :: Id Ilk,           Urn type  
  · lad :: Entity,          Urn owner  
  · art :: Wad,             Stablecoin issuance in fee unit  
  · ink :: Wad,             Amount of locked assetcoin  
  · cat :: Maybe Entity     Entity that triggered riddance, if applicable  
} deriving (Eq, Show)
```

### 3.7 Vox — feedback mechanism data

The *feedback mechanism* is the aspect of the system that adjusts the target price of dai based on market price. Its data is grouped in a record called Vox.

```
data Vox = Vox {  
  · wut :: Wad,   Stablecoin market price denominated in sdr  
  · par :: Wad,   Stablecoin target price denominated in sdr  
  · way :: Ray,   Current per-second change in target price  
  · how :: Ray,   Sensitivity parameter set by governance  
  · tau :: Sec    Timestamp of latest feedback iteration  
} deriving (Eq, Show)
```

### 3.8 Vat — system root

The Vat record aggregates the records of tokens, urns, ilks, and price feeds, along with the data of the feedback mechanism.

```
data Vat = Vat {  
  · tags :: Map Gem Tag,      Assetcoin price feeds
```

- `ilks :: Map (Id Ilk) Ilk`, Urn type records
- `urns :: Map (Id Urn) Urn`, Urn records
- `vox :: Vox` Feedback mechanism data

} **deriving** (Eq, Show)

## 3.9 System model

Finally we define a record with no direct counterpart in the Solidity contracts, which has the Vat record along with model state.

```
data System = System {
  · balances :: Map (Entity, Gem) Wad, Token balances
  · vat      :: Vat, Root Maker entity
  · era      :: Sec, Current time stamp
  · sender   :: Entity, Sender of current act
  · accounts :: [Address], For test suites
  · mode     :: Mode Vow operation mode
} deriving (Eq, Show)
```

```
data Mode = Dummy
  deriving (Eq, Show)
```

## 3.10 Default data

```
defaultIlk :: Gem → Ilk
defaultIlk idgem = Ilk {
  · gem = idgem,
  · axe = Ray 1,
  · mat = Ray 1,
  · tax = Ray 1,
  · hat = Wad 0,
  · lax = Sec 0,
  · chi = Ray 1,
  · rum = Wad 0,
```

```

    · rho = Sec 0
  }

emptyUrn :: Id Ilk → Entity → Urn
emptyUrn idilk idlad = Urn {
  · cat = Nothing,
  · lad = idlad,
  · ilk = idilk,
  · art = Wad 0,
  · ink = Wad 0
}

initialTag :: Tag
initialTag = Tag {
  · tag = Wad 0,
  · zzz = 0
}

initialVat :: Ray → Vat
initialVat how0 = Vat {
  · vox = Vox {
    · tau = 0,
    · wut = Wad 1,
    · par = Wad 1,
    · how = how0,
    · way = Ray 1
  },
  · ilks = ∅,
  · urns = ∅,
  · tags = ∅
}

initialSystem :: Ray → System
initialSystem how0 = System {
  · balances = ∅,
  · vat = initialVat how0,
  · era = 0,
  · sender = God,
  · accounts = mempty,
  · mode = Dummy
}

```

# Chapter 4

## Acts

The *acts* are the basic state transitions of the system.

Unless specified as *internal*, acts are accessible as public functions on the blockchain.

The `auth` modifier marks acts which can only be invoked from addresses to which the system has granted authority.

For details on the underlying “Maker monad,” which specifies how the act definitions behave with regard to state and rollback, see [chapter 5](#).



## 4.1 Assessment

In order to prohibit urn acts based on risk situation, we define these stages of risk.

```
data Stage = Pride | Anger | Worry | Panic | Grief | Dread
deriving (Eq, Show)
```

We define the function *analyze* that determines the risk stage of an urn.

```
analyze era0 par0 urn0 ilk0 tag0 =
  if | view cat urn0 ≠ Nothing ∧ view ink urn0 ≡ 0
    Riddance triggered and started
    → Dread
  | view cat urn0 ≠ Nothing
    Riddance triggered
    → Grief
  | pro < min
    Locked assetcoin value below issuance times riddance ratio
    → Panic
  | view zzz tag0 + view lax ilk0 < era0
    Assetcoin price limbo exceeded limit
    → Panic
  | view zzz tag0 < era0
    Assetcoin price feed in limbo
    → Worry
  | cap > view hat ilk0
    Issuance ceiling exceeded
    → Anger
  | otherwise
    No problems
    → Pride
```

**where**

Value of urn's locked assetcoin in SDR:

*pro* = *view ink urn*<sub>0</sub> \* *view tag tag*<sub>0</sub>

Ilk's total stablecoin issuance in DAI:

*cap* = *view rum ilk*<sub>0</sub> \* *cast (view chi ilk*<sub>0</sub>)

























Urn's stablecoin issuance denominated in SDR:






*con* = *view art urn*<sub>0</sub> \* *cast (view chi ilk*<sub>0</sub>) \* *par*<sub>0</sub>

Required assetcoin value as per riddance ratio:

*min* = *con* \* *cast (view mat ilk*<sub>0</sub>)

Table 4.1: Urn acts and risk stages

|       | give  | shut  | lock  | wipe  | free  | draw  | bite  | grab  | plop  |
|-------|---|---|---|---|---|---|---|---|---|
| Pride |  |  |  |  |  |  | —   | —   | —   |
| Anger |  |  |  |  |  | —   | —   | —   | —   |
| Worry |  |  |  |  | —   | —   | —   | —   | —   |
| Panic |  |  |  |  | —   | —   |  | —   | —   |
| Grief |  | —   | —   | —   | —   | —   | —   |  | —   |
| Dread |  | —   | —   | —   | —   | —   | —   | —   |  |
|       | decrease risk   |   |   | increase risk   |   |   | unwind risk   |   |   |

-  allowed for anyone
-  allowed for owner unconditionally
-  allowed for owner if able to pay
-  allowed for owner if above riddance ratio
-  allowed for settler contract

Now we define the internal act `feel` which returns the value of *analyze* after ensuring that the system state is updated.

```

feel idurn = do
  Adjust target price and target rate
  prod
  Update fee unit and unprocessed fee revenue
  idilk ← look (vat . urns . ix idurn . ilk)
  drip idilk
  Read parameters for risk analysis
  era0 ← use era
  par0 ← use (vat . vox . par)
  urn0 ← look (vat . urns . ix idurn)
  ilk0 ← look (vat . ilks . ix (view ilk urn0))
  tag0 ← look (vat . tags . ix (view gem ilk0))
  Return risk stage of urn
  return (analyze era0 par0 urn0 ilk0 tag0)

```

Urn acts use `feel` to prohibit increasing risk when already risky, and to freeze stablecoin and assetcoin during riddance; see Table 4.1.

## 4.2 Issuance

Any user can open one or more accounts with the system using `open`, specifying a self-chosen account identifier and an ilk.

```
open  $id_{urn}$   $id_{ilk}$  = do  
  Fail if account identifier is taken  
     $none$  ( $vat . urns . ix\ id_{urn}$ )  
  Fail if ilk type is not present  
     $\_ \leftarrow look$  ( $vat . ilks . ix\ id_{ilk}$ )  
  Create an urn with the sender as owner  
     $id_{lad} \leftarrow use\ sender$   
     $initialize$  ( $vat . urns . at\ id_{urn}$ ) ( $emptyUrn\ id_{ilk}\ id_{lad}$ )
```

The owner of an urn can transfer its ownership at any time using `give`.

```
give  $id_{urn}$   $id_{lad}$  = do  
  Fail if sender is not the urn owner  
     $id_{sender} \leftarrow use\ sender$   
     $owns\ id_{urn}\ id_{sender}$   
  Transfer urn ownership  
     $vat . urns . ix\ id_{urn} . lad := id_{lad}$ 
```

Unless urn is in riddance, its owner can use `lock` to lock more assetcoin.

```
lock  $id_{urn}$   $wad_{gem}$  = do  
  Fail if sender is not the urn owner  
     $id_{lad} \leftarrow use\ sender$   
     $owns\ id_{urn}\ id_{lad}$   
  Fail if riddance in process  
     $want$  ( $feel\ id_{urn}$ ) ( $\notin [Grief, Dread]$ )  
  Identify assetcoin  
     $id_{ilk} \leftarrow look$  ( $vat . urns . ix\ id_{urn} . ilk$ )  
     $id_{gem} \leftarrow look$  ( $vat . ilks . ix\ id_{ilk} . gem$ )  
  Take custody of assetcoin  
     $transfer\ id_{gem}\ wad_{gem}\ id_{lad}\ Jar$   
  Record an assetcoin balance increase  
     $increase$  ( $vat . urns . ix\ id_{urn} . ink$ )  $wad_{gem}$ 
```

When an urn has no risk problems (except that its ilk's ceiling may be exceeded), its owner can use free to reclaim some amount of assetcoin, as long as this would not take the urn below its riddance ratio.

```

free  $id_{urn}$   $wad_{gem} = \mathbf{do}$ 
  Fail if sender is not the urn owner
   $id_{lad} \leftarrow use\ sender$ 
   $owns\ id_{urn}\ id_{lad}$ 
  Record an assetcoin balance decrease
   $decrease\ (vat.\ urns.\ ix\ id_{urn}.\ ink)\ wad_{gem}$ 
  Roll back on any risk problem except ilk ceiling excess
   $want\ (feel\ id_{urn})\ (\in [Pride, Anger])$ 
  Release custody of assetcoin quantity
   $id_{ilk} \leftarrow look\ (vat.\ urns.\ ix\ id_{urn}.\ ilk)$ 
   $id_{gem} \leftarrow look\ (vat.\ ilks.\ ix\ id_{ilk}.\ gem)$ 
   $transfer\ id_{gem}\ wad_{gem}\ Jar\ id_{lad}$ 

```

When an urn has no risk problems, its owner can use draw to issue fresh stablecoin, as long as the ilk ceiling is not exceeded and the issuance would not take the urn below its riddance ratio.

```

draw  $id_{urn}$   $wad_{DAI} = \mathbf{do}$ 
  Fail if sender is not the urn owner
   $id_{lad} \leftarrow use\ sender$ 
   $owns\ id_{urn}\ id_{lad}$ 
  Update fee unit and unprocessed fee revenue
   $id_{ilk} \leftarrow look\ (vat.\ urns.\ ix\ id_{urn}.\ ilk)$ 
   $chi_1 \leftarrow drip\ id_{ilk}$ 
  Denominate issuance quantity in fee unit
  let  $wad_{chi} = wad_{DAI} / cast\ chi_1$ 
  Record increase of urn's stablecoin issuance
   $increase\ (vat.\ urns.\ ix\ id_{urn}.\ art)\ wad_{chi}$ 
  Record increase of ilk's stablecoin issuance
   $increase\ (vat.\ ilks.\ ix\ id_{ilk}.\ rum)\ wad_{chi}$ 
  Roll back on any risk problem
   $want\ (feel\ id_{urn})\ (\equiv Pride)$ 
  Mint both stablecoin and anticon
  lend  $wad_{DAI}$ 
  Transfer stablecoin to urn owner

```

*transfer* DAI  $wad_{DAI}$  Joy  $id_{lad}$

Transfer anticon to anticon holding account

*transfer* SIN  $wad_{DAI}$  Woe Ice

An urn owner who has previously issued stablecoin can use *wipe* to send back dai and reduce the urn's issuance.

*wipe*  $id_{urn}$   $wad_{DAI}$  = **do**

Fail if sender is not the urn owner

$id_{lad} \leftarrow use\ sender$

*owns*  $id_{urn}$   $id_{lad}$

Fail if urn is in riddance

*want* (*feel*  $id_{urn}$ ) ( $\notin$  [Grief, Dread])

Update fee unit and unprocessed fee revenue

$id_{ilk} \leftarrow look\ (vat.urns.ix\ id_{urn}.ilk)$

$chi_1 \leftarrow drip\ id_{ilk}$

Denominate stablecoin amount in fee unit

**let**  $wad_{chi} = wad_{DAI} / cast\ chi_1$

Record decrease of urn issuance

*decrease* ( $vat.urns.ix\ id_{urn}.art$ )  $wad_{chi}$

Record decrease of ilk total issuance

*decrease* ( $vat.ilks.ix\ id_{ilk}.rum$ )  $wad_{chi}$

Take custody of stablecoin from urn owner

*transfer* DAI  $wad_{DAI}$   $id_{lad}$  Jar

Destroy stablecoin and anticon

*mend*  $wad_{DAI}$

An urn owner can use *shut* to close their account—reversing all issuance plus fee and reclaiming all assetcoin—if the price feed is up to date and the urn is not in riddance.

*shut*  $id_{urn}$  = **do**

Update fee unit and unprocessed fee revenue

$id_{ilk} \leftarrow look\ (vat.urns.ix\ id_{urn}.ilk)$

$chi_1 \leftarrow drip\ id_{ilk}$

Reverse all issued stablecoin plus fee

$art_0 \leftarrow look\ (vat.urns.ix\ id_{urn}.art)$

*wipe*  $id_{urn}$  ( $art_0 * cast\ chi_1$ )

Reclaim all locked assetcoin

$ink_0 \leftarrow look\ (vat.urns.ix\ id_{urn}.ink)$

```
    free  $id_{\text{urn}}$  ink0  
Nullify urn record  
    vat . urns . at  $id_{\text{urn}}$  := Nothing
```

## 4.3 Adjustment

The feedback mechanism is updated through `prod`, which can be invoked at any time by keepers, but is also invoked as a side effect of any urn act that uses `feel` to assess risk.

```
prod = do
  Read all parameters relevant for feedback mechanism
  era0 ← use era
  tau0 ← use (vat . vox . tau)
  wut0 ← use (vat . vox . wut)
  par0 ← use (vat . vox . par)
  how0 ← use (vat . vox . how)
  way0 ← use (vat . vox . way)
let
  Time difference in seconds
  age = era0 − tau0
  Current target rate applied to target price
  par1 = par0 * cast (way0 ↑↑ age)
  Sensitivity parameter applied over time
  wag = how0 * fromIntegral age
  Target rate scaled up or down
  way1 = inj (prj way0 +
    if wut0 < par0 then wag else − wag)
  Update target price
  vat . vox . par := par1
  Update rate of price change
  vat . vox . way := way1
  Record time of update
  vat . vox . tau := era0
where
  Convert between multiplicative and additive form
  prj x = if x ≥ 1 then x − 1 else 1 − 1 / x
  inj x = if x ≥ 0 then x + 1 else 1 / (1 − x)
```

The stability fee of an ilk can change through governance. Due to the constraint that acts should run in constant time, the system cannot iterate over urns to effect such changes. Instead each ilk has a single “fee unit” which accumulates the stability fee. The drip act updates this unit. It can be called at any time by keepers, but is also called as a side effect of every act that uses `feel` to assess urn risk.

```

drip  $id_{ilk}$  = do
  Time stamp of previous drip
   $\rho_0 \leftarrow look(vat.ilks.ix\ id_{ilk}.\rho)$ 
  Current stability fee
   $\tau_0 \leftarrow look(vat.ilks.ix\ id_{ilk}.\tau)$ 
  Current fee unit value
   $\chi_0 \leftarrow look(vat.ilks.ix\ id_{ilk}.\chi)$ 
  Current total issuance in fee unit
   $\text{rum}_0 \leftarrow look(vat.ilks.ix\ id_{ilk}.\text{rum})$ 
  Current time stamp
   $\text{era}_0 \leftarrow use\ era$ 
  let
    Time difference in seconds
     $age = \text{era}_0 - \rho_0$ 
    Value of fee unit increased according to stability fee
     $\chi_1 = \chi_0 * \tau_0 \uparrow\uparrow age$ 
    Stability fee revenue denominated in new unit
     $dew = (cast(\chi_1 - \chi_0) :: Wad) * \text{rum}_0$ 
  Mint stablecoin and anticon for marginally accrued fee
  lend  $dew$ 
  Record time of update
   $vat.ilks.ix\ id_{ilk}.\rho := \text{era}_0$ 
  Record new fee unit
   $vat.ilks.ix\ id_{ilk}.\chi := \chi_1$ 
  Return the new fee unit
  return  $\chi_1$ 

```



## 4.4 Price feed input

The `mark` act records a new market price of an assetcoin along with the expiration date of this price.

```
mark  $id_{gem}$  tag1 zzz1 = auth $ do  
  initialize (vat . tags . at  $id_{gem}$ ) Tag {  
    · tag = tag1,  
    · zzz = zzz1  
  }
```

The `tell` act records a new market price of the DAI token along with the expiration date of this price.

```
tell wad = auth $ do vat . vox . wut := wad
```

## 4.5 Riddance

When an urn's stage marks it as in need of riddance, any account can invoke the `bite` act to trigger the riddance process. This enables the settler contract to grab the assetcoin for auctioning and take over the anticoin.

```
bite  $id_{urn}$  = do  
  Fail if urn is not in the appropriate stage  
  want (feel  $id_{urn}$ ) ( $\equiv$  Panic)  
  Record the sender as the riddance initiator  
   $id_{cat} \leftarrow use\ sender$   
  vat . urns . ix  $id_{urn}$  . cat := Just  $id_{cat}$   
  Apply riddance penalty to urn issuance  
   $id_{ilk} \leftarrow look\ (vat . urns . ix\ id_{urn} . ilk)$   
   $axe_0 \leftarrow look\ (vat . ilks . ix\ id_{ilk} . axe)$   
   $art_0 \leftarrow look\ (vat . urns . ix\ id_{urn} . art)$   
  let  $art_1 = art_0 * cast\ axe_0$   
  Update urn issuance to include penalty  
  vat . urns . ix  $id_{urn}$  . art :=  $art_1$ 
```

After riddance has been triggered, the designated settler contract invokes `grab` to receive both the urn's assetcoin and the anticones corresponding to the urn's issuance.

```

grab  $id_{urn}$  = auth $ do
  Fail if urn is not marked for riddance
  want (feel  $id_{urn}$ ) ( $\equiv$  Grief)
   $ink_0 \leftarrow look$  (vat . urns . ix  $id_{urn}$  . ink)
   $art_0 \leftarrow look$  (vat . urns . ix  $id_{urn}$  . art)
   $id_{ilk} \leftarrow look$  (vat . urns . ix  $id_{urn}$  . ilk)
   $id_{gem} \leftarrow look$  (vat . ilks . ix  $id_{ilk}$  . gem)
  Update the fee unit and unprocessed fee revenue
   $chi_1 \leftarrow drip$   $id_{ilk}$ 
  Denominate the issuance in dai
  let con =  $art_0 * cast$   $chi_1$ 
  Transfer assetcoin and anticon to settler
  transfer  $id_{gem}$   $ink_0$  Jar Vow
  transfer SIN con Jar Vow
  Nullify urn's assetcoin and anticon quantities
  vat . urns . ix  $id_{urn}$  . ink := 0
  vat . urns . ix  $id_{urn}$  . art := 0
  Decrease the ilk's total issuance
  decrease (vat . ilks . ix  $id_{ilk}$  . rum)  $art_0$ 

```

When the settler has finished the riddance of an urn, it invokes `plop` to give back any assetcoin it did not need to sell and restore the urn.

```

plop  $id_{urn}$  wadDAI = auth $ do
  Fail unless urn is in the proper stage
  want (feel  $id_{urn}$ ) ( $\equiv$  Dread)
  Forget the urn's initiator of riddance
  vat . urns . ix  $id_{urn}$  . cat := Nothing
  Take excess assetcoin from settler to vault
   $id_{vow} \leftarrow use$  sender
   $id_{ilk} \leftarrow look$  (vat . urns . ix  $id_{urn}$  . ilk)
   $id_{gem} \leftarrow look$  (vat . ilks . ix  $id_{ilk}$  . gem)
  transfer  $id_{gem}$  wadDAI  $id_{vow}$  Jar
  Record the excess assetcoin as belonging to the urn
  vat . urns . ix  $id_{urn}$  . ink := wadDAI

```

The settler can invoke `loot` at any time to claim all uncollected stability fee revenue for use in the countercoin buy-and-burn auction.

```
loot = auth $ do
```

```
  The dai vault's balance is the uncollected stability fee revenue
```

```
  wad ← look (balance DAI Jar)
```

```
  Transfer the entire dai vault balance to sender
```

```
   $id_{\text{vow}} \leftarrow \textit{use sender}$ 
```

```
  transfer DAI wad Jar Vow
```

## 4.6 Auctioning

```
flip  $id_{\text{gem}}$  wad_jam wad_tab  $id_{\text{urn}}$  = do
```

```
  vow ← look mode
```

```
  case vow of
```

```
    Dummy → return ()
```

```
flap = do
```

```
  vow ← look mode
```

```
  case vow of
```

```
    Dummy → return ()
```

```
flop = do
```

```
  vow ← look mode
```

```
  case vow of
```

```
    Dummy → return ()
```

## 4.7 Settlement

```
tidy who = auth $ do
  Find the entity's stablecoin and anticon balances
  awe ← look (balance DAI who)
  woe ← look (balance SIN who)
  We can burn at most the smallest of the two balances
  let x = min awe woe
  Transfer stablecoin and anticon to the settler
  transfer DAI x who Vow
  transfer SIN x who Vow
  Burn both stablecoin and anticon
  burn   DAI x      Vow
  burn   SIN x      Vow

kick = do
  Transfer unprocessed stability fee revenue to vow account
  loot
  Cancel stablecoin against anticon
  tidy Vow
  Assign any remaining stablecoin to countercoin-deflating auction
  transferAll DAI Vow Flapper
  flap
  Assign any remaining anticon to countercoin-inflating auction
  transferAll SIN Vow Flopper
  flop
```

## 4.8 Governance

Governance uses `form` to create a new ilk. Since the new type is initialized with a zero ceiling, a separate transaction can safely set the risk parameters before any issuance occurs.

```
form idilk idgem = auth $ do
  initialize (vat . ilks . at idilk) (defaultIlk idgem)
```

Governance uses `frob` to alter the sensitivity factor, which is the only mutable parameter of the feedback mechanism.

```
frob how1 = auth $ do vat.vox.how := how1
```

Governance can alter the five risk parameters of an ilk using `cuff` for the riddance ratio; `chop` for the riddance penalty; `cork` for the ilk ceiling; `calm` for the duration of price limbo; and `crop` for the stability fee.

```
cuff idilk mat1 = auth $ do vat.ilks.ix idilk.mat := mat1
chop idilk axe1 = auth $ do vat.ilks.ix idilk.axe := axe1
cork idilk hat1 = auth $ do vat.ilks.ix idilk.hat := hat1
calm idilk lax1 = auth $ do vat.ilks.ix idilk.lax := lax1
```

When altering the stability fee with `crop`, we ensure that the previous stability fee has been accounted for in the internal fee unit.

```
crop idilk tax1 = auth $ do
  Apply the current stability fee to the internal fee unit
  drip idilk
  Change the stability fee
  vat.ilks.ix idilk.tax := tax1
```

## 4.9 Token manipulation

We model the ERC20 transfer function in simplified form (omitting the concept of “allowance”).

```
transfer idgem wad src dst =
  Operate in the token’s balance table
  zoom balances $ do
    Fail if source balance insufficient
    balance ← look (ix (src, idgem))
    aver (balance ≥ wad)
  Update balances
  decrease (ix (src, idgem)) wad
  initialize (at (dst, idgem)) 0
  increase (ix (dst, idgem)) wad
```

```

transferAll  $id_{gem}$   $src$   $dst = do$ 
   $wad \leftarrow look (balance\ id_{gem}\ src)$ 
  transfer  $id_{gem}$   $wad$   $src$   $dst$ 

```

The internal act *mint* inflates the supply of a token. It is used by *lend* to create new stablecoin and anticon, and by the settler to create new countercoin.

```

mint  $id_{gem}$   $wad$   $dst = do$ 
  initialize ( $balances . at\ (dst, id_{gem})$ ) 0
  increase ( $balances . ix\ (dst, id_{gem})$ )  $wad$ 

```

The internal act *burn* deflates the supply of a token. It is used by *mend* to destroy stablecoin and anticon, and by the settler to destroy countercoin.

```

burn  $id_{gem}$   $wad$   $src =$ 
  decrease ( $balances . ix\ (src, id_{gem})$ )  $wad$ 

```

The internal act *lend* mints identical amounts of both stablecoin and anticon. It is used by *draw* to issue stablecoin; it is also used by *drip* to issue stablecoin representing revenue from stability fees, which stays in the vault until collected.

```

lend  $wad_{DAI} = do$ 
  mint DAI  $wad_{DAI}$  Joy
  mint SIN  $wad_{DAI}$  Woe

```

The internal act *mend* destroys identical amounts of both dai and the internal debt token. Its use via *wipe* is how the stablecoin supply is reduced.

```

mend  $wad_{DAI} = do$ 
  burn DAI  $wad_{DAI}$  Jar
  burn SIN  $wad_{DAI}$  Ice

```

# Chapter 5

## Act framework

The reader does not need any abstract understanding of monads to understand the code. They give us a nice syntax—the **do** block notation—for expressing exceptions and state in a way that is still purely functional. Each line of such a block is interpreted by the monad to provide the semantics we want.

### 5.1 The Maker monad

This defines the Maker monad as a simple composition of a state monad and an error monad:

```
type Maker a = StateT System (Except Error) a
```

We divide act failure modes into general assertion failures and authentication failures.

```
data Error = AssertError Act | AuthError
  deriving (Show, Eq)
```

An act can be executed on a given initial system state using *exec*. The result is either an error or a new state. The *exec* function can also accept a sequence of acts, which will be interpreted as a single transaction.

```
exec :: System → Maker () → Either Error System
exec sys m = runExcept (execStateT m sys)
```

## 5.2 Asserting

We now define a set of functions that fail unless some condition holds.

General assertion  
 $\text{aver } x = \text{unless } x \text{ (throwError (AssertError ?act))}$   
Assert that an indexed value is not present  
 $\text{none } x = \text{preuse } x \gg= \lambda \text{case}$   
    Nothing  $\rightarrow \text{return } ()$   
    Just  $\_ \rightarrow \text{throwError (AssertError ?act)}$   
Assert that an indexed value is present  
 $\text{look } f = \text{preuse } f \gg= \lambda \text{case}$   
    Nothing  $\rightarrow \text{throwError (AssertError ?act)}$   
    Just  $x \rightarrow \text{return } x$   
Execute an act and assert a condition on its result  
 $\text{want } m \text{ } p = m \gg= (\text{aver } . p)$

We define  $\text{owns } id_{\text{urn}} \text{ } id_{\text{lad}}$  as an assertion that the given CDP is owned by the given account.

$\text{owns } id_{\text{urn}} \text{ } id_{\text{lad}} = \text{do}$   
     $\text{want (look (vat . urns . ix } id_{\text{urn}} \text{ . lad)) } (\equiv id_{\text{lad}})$

We define  $\text{auth } k$  as an act modifier that executes  $k$  only if the sender is authorized.

$\text{auth } \text{continue} = \text{do}$   
     $s \leftarrow \text{use sender}$   
     $\text{unless } (s \equiv \text{God}) \text{ (throwError AuthError)}$   
     $\text{continue}$



# Appendix A

## Prelude

This module reexports symbols from other packages and exports a few new symbols of its own.

```
module Maker.Prelude (module Maker.Prelude, module X) where
import Prelude as X (
  Conversions to and from strings
    Read (.), Show (.), read,
  Comparisons
    Eq (.), Ord (.),
  Core abstractions
    Functor    (fmap),
    Applicative (),
    Monad      (return, (>>=)),
  Numeric classes
    Num (.), Integral (), Enum (),
  Numeric conversions
    Real (.), Fractional (.),
    RealFrac (.),
    fromIntegral,
  Simple types
    Integer, Int, String,
  Algebraic types
    Bool    (True, False),
    Maybe (Just, Nothing),
    Either  (Right, Left),
```

Functional operators  
 ( $\cdot$ ), ( $\$$ ),  
 Numeric operators  
 ( $+$ ), ( $-$ ), ( $*$ ), ( $/$ ), ( $\uparrow$ ), ( $\uparrow\uparrow$ ), *div*,  
 Utilities  
*all*,  $\neg$ , *elem*, ( $\wedge$ ),  
 Constants  
*mempty*,  $\perp$ , *otherwise*)

We use a typical composition of monad transformers from the `mtl` library to structure stateful actions. See section 5.1 (*The Maker monad*).

```
import Control.Monad.State as X (
  StateT,      Type constructor that adds state to a monad type
  execStateT,  Runs a state monad with given initial state
  get,         Gets the state in a do block
  put)         Sets the state in a do block
import Control.Monad.Writer as X (
  WriterT,    Type constructor that adds logging to a monad type
  Writer,     Type constructor of logging monads
  runWriterT, Runs a writer monad transformer
  execWriterT, Runs a writer monad transformer keeping only logs
  execWriter) Runs a writer monad keeping only logs
import Control.Monad.Except as X (
  MonadError, Type class of monads that fail
  Except,     Type constructor of failing monads
  throwError, Short-circuits the monadic computation
  runExcept)  Runs a failing monad
```

Our numeric types use decimal fixed-point arithmetic.

```
import Data.Fixed as X (
  Fixed (.),      Type constructor for numbers of given precision
  HasResolution (..)) Type class for specifying precisions
```

We rely on the `lens` library for accessing nested values. There is no need to understand the theory behind lenses to understand this program. The notation  $a . b . c$  denotes a nested accessor much like `a.b.c` in C-style languages; for more details, consult `lens` documentation<sup>1</sup>.

---

<sup>1</sup>Gabriel Gonzalez's 2013 article *Program imperatively using Haskell* is a good introduction.

```

import Control.Lens as X (
  Lens', lens,
  makeLenses,  Defines lenses for record fields
  makeFields,  Defines lenses for record fields
  set,         Writes a lens
  use, preuse, Reads a lens from a state value
  view,        Reads a lens from a value
  ix,          Lens for map retrieval and updating
  at,          Lens for map insertion

  Operators for partial state updates in do blocks:
  (:=),        Replace
  (-=), (+=),  Update arithmetically
  (%=),        Update according to function
  (?=))        Insert into map

import Control.Lens.Zoom as X (zoom)

```

Where the Solidity code uses mapping, we use Haskell's regular tree-based map type<sup>2</sup>.

```

import Data.Map as X (
  Map,      Type constructor for mappings
  ∅,        Polymorphic empty mapping
  singleton, Creates a mapping with a single key-value pair
  fromList) Creates a mapping with several key-value pairs

```

Finally we define some of our own convenience functions.

```

decrease a x = a -= x
increase a x = a += x
initialize a x = a %= (λcase Nothing → Just x; y → y)
prepend a x = a %= (x:)
x ∉ xs = ¬ (elem x xs)

```

---

<sup>2</sup>We assume the axiom that Keccak hash collisions are impossible.

## Appendix B

### Fixed point numbers with rounding

This somewhat arcane-looking code implements a wrapper around the base library's decimal fixed point type, only with  $x * y$  and  $x / y$  operations that do rounding instead of truncation of their intermediate results.

```
module Maker.Decimal (Decimal (.), E18, E36, Epsilon (. .)) where  
import Data.Fixed  
newtype HasResolution  $e \Rightarrow$  Decimal  $e =$  D (Fixed  $e$ )  
  deriving (Ord, Eq, Real, RealFrac)
```

We want the printed representations of these numbers to look like "0.01" and not "R 0.01".

```
instance HasResolution  $e \Rightarrow$  Read (Decimal  $e$ ) where  
  readsPrec  $n s = fmap (\lambda(x, y) \rightarrow (D x, y)) (readsPrec n s)  
instance HasResolution  $e \Rightarrow$  Show (Decimal  $e$ ) where  
  show (D  $x$ ) = show  $x$$ 
```

In the Num instance, we delegate everything except multiplication.

```
instance HasResolution  $e \Rightarrow$  Num (Decimal  $e$ ) where  
   $x@(D (MkFixed a)) * D (MkFixed b) =$   
    D (MkFixed (div ( $a * b + \text{div } (resolution\ x) 2$ )  
                  (resolution  $x$ )))  
  D  $a + D b = D (a + b)$   
  D  $a - D b = D (a - b)$   
  negate (D  $a$ ) = D (negate  $a$ )  
  abs (D  $a$ ) = D (abs  $a$ )
```

```

signum (D a) = D (signum a)
fromInteger i = D (fromInteger i)

```

In the Fractional instance, we delegate everything except division.

```

instance HasResolution e  $\Rightarrow$  Fractional (Decimal e) where
  x@(D (MkFixed a)) / D (MkFixed b) =
    D (MkFixed (div (a * resolution x + div b 2) b))
  recip (D a)      = D (recip a)
  fromRational r = D (fromRational r)

```

We define the E18 and E36 symbols and their fixed point multipliers.

```

data E18; data E36
instance HasResolution E18 where
  resolution _ = 10  $\uparrow$  (18 :: Integer)
instance HasResolution E36 where
  resolution _ = 10  $\uparrow$  (36 :: Integer)

```

The fixed point number types have well-defined smallest increments (denoted  $\epsilon$ ). This becomes useful when verifying equivalences.

```

class Epsilon t where  $\epsilon :: t$ 
instance HasResolution a  $\Rightarrow$  Epsilon (Decimal a) where
  The use of  $\perp$  is safe since resolution ignores the value.
   $\epsilon = 1 / \text{fromIntegral } (\text{resolution } (\perp :: \text{Fixed } a))$ 

```