



presents the

REFERENCE IMPLEMENTATION

of the remarkable

DAI CREDIT SYSTEM

issuing a diversely collateralized stablecoin

with last update on March 6, 2017.

Contents

1	Introduction	5
1.1	Reference implementation	6
1.2	Prerequisite Haskell knowledge	7
I	Implementation	9
2	Preamble	11
3	Types	13
3.1	Numeric types	13
3.2	Identifier type	14
3.3	Domain types	14
	Address — Ethereum accounts	14
	Gem — ERC20 token model	15
	Jar — collateral type	15
	Ilk — CDP type	15
	Urn — collateralized debt position (CDP)	16
	Vat — CDP engine	16
	System model	17
3.4	Default data	17
4	Act framework	21
4.1	Act descriptions	21
4.2	The Maker monad	22
4.3	Constraints	23
4.4	Accessor aliases	23
4.5	Logging and asserting	23
4.6	Modifiers	23
	note — logging actions	23
	auth — authenticating actions	24

5	Acts	25
5.1	Risk assessment	26
	gaze — identify CDP risk stage	26
5.2	Lending	27
	open — create CDP account	27
	lock — deposit collateral	27
	free — withdraw collateral	27
	draw — issue dai as debt	28
	wipe — repay debt and burn dai	28
	give — transfer CDP account	29
	shut — wipe, free, and delete CDP	29
5.3	Frequent adjustments	31
	prod — perform revaluation and rate adjustment	31
	drip — update price of debt coin	31
5.4	Governance	32
	form — create a new CDP type	32
	frob — set the sensitivity parameter	32
5.5	Price feedback	32
	mark — update market price of dai	32
	tell — update market price of collateral token	32
5.6	Liquidation and settlement	33
	bite — mark CDP for liquidation	33
	grab — take tokens to begin CDP liquidation	33
	heal — process bad debt	33
	loot — process stability fee revenue	34
5.7	Minting, burning, and transferring	34
	pull — take tokens to vat	34
	push — send tokens from vat	34
	mint — increase supply	34
	burn — decrease supply	34
5.8	Test-related manipulation	34
	warp — travel in time	34
5.9	Other stuff	35
6	Testing	37
A	Act type signatures	39

Chapter 1

Introduction

The DAI CREDIT SYSTEM, henceforth also “Maker,” is a network of Ethereum contracts designed to issue the DAI currency token and automatically adjust incentives in order to keep dai market value stable relative to SDR¹ in the short and medium term.

New dai enters the money supply when a borrower takes out a loan backed by an excess of collateral locked in Maker’s token vault. The debt and collateral amounts are recorded in a *collateralized debt position*, or CDP. Thus all outstanding dai represents some CDP owner’s claim on their collateral.

Maker’s knowledge of the market values of dai and the various tokens used as collateral comes from *price feeds*. Prices are used to continuously assess the risk of each CDP. If the value of a CDP’s collateral drops below a certain multiple of its debt, it is marked for liquidation, which triggers a decentralized auction mechanism.

Another token, MKR, is also controlled by Maker, acting as a “share” in the system itself. When a CDP liquidation fails to recover the full value of debt, Maker mints more MKR and auctions it out. Thus MKR is used to fund last resort market making. The value of the MKR token is based on the *stability fee* imposed on all dai loans: stability fee revenue goes toward buying MKR for burning.

This document is an executable technical specification of the exact workings of the Maker smart contracts.

¹“Special Drawing Rights” (ticker symbol XDR), the international reserve asset created by the International Monetary Fund, whose value is derived from a weighted basket of world currencies. In the long term, the value of dai may diverge from the value of SDR; whether in an inflationary or deflationary way will depend on market forces.

1.1 Reference implementation

The version of this system that will be deployed on the Ethereum blockchain is written in Solidity, which is a workable smart contract implementation language. This reference implementation is a precise model of the behavior of those contracts, written as a “literate” Haskell program. The motivations for such a reference implementation include:

1. **Comparison.** Checking two free-standing implementations against each other is a well-known way of ensuring that they both behave as intended.
2. **Testing.** Haskell lets us use flexible and powerful testing tools such as QuickCheck and SmallCheck for comprehensively verifying key properties as a middle ground between unit testing and formal verification.
3. **Explicitness.** Coding the contract behavior in Haskell, a purely functional language, enforces explicit description of aspects which Solidity leaves implicit. For example, a Solidity program can read a previously unwritten mapping and get back a value initialized with zeroed memory, whereas in Haskell we must explicitly describe default values. The state rollback behavior of failed actions is also in Haskell explicitly coded as part of the monad transformer stack.
4. **Type correctness.** While Solidity does have a static type system, it is not expressive enough to encode the distinctions made by our system. In particular, the two different decimal fixed point number types that we use are typed in Solidity with one and the same `uint128` type. In Haskell we can make this distinction explicit.
5. **Formality.** The work of translating a Solidity program into a purely functional program opens up opportunities for certain types of formal verification. In particular, this document will be useful for modelling aspects of the system in a proof assistant like Agda, Idris, Coq, or Isabelle. We can also use logical tools for Haskell, such as Liquid Haskell (which provides compile time logical property checking) and `sbv` (a toolkit for model checking and symbolic execution).
6. **Simulation.** Solidity is highly specific to the Ethereum blockchain environment and as such does not have facilities for interfacing with files or other computer programs. This makes the Solidity implementation of the system less useful for doing simulations of the system’s economic, game-theoretic, or statistical aspects.

1.2 Prerequisite Haskell knowledge

Some parts of this document require specific knowledge about Haskell programming, but these parts only make up a framework for expressing the more interesting parts in a natural way free of boilerplate.

◇ *Guidelines for skipping boring chapters and so on...*

For a complete understanding of the reference implementation’s source code, the reader should grasp the following Haskell patterns:

- The use of **newtype** wrappers to distinguish different types of values which have the same underlying type.
- The use of **do** notation with the standard monad transformers:
 - **StateT** for updating state,
 - **ReaderT** for the read-only environment,
 - **WriterT** for “write-only state” (namely logs), and
 - **ExceptT** for failures which roll back state changes.
- The basic use of “lenses” (via the **lens** library) for convenient reading and writing of specific parts of nested values.
- The use of “parametricity” to express type-level guarantees about how function parameters are used, especially for understanding [Appendix A](#) which uses type signatures to specify which parts of the system are used or altered by each system action.
- ◇ *Some more stuff here...*

Part I

Implementation

Chapter 2

Preamble

We declare the program’s dependencies up front. The reader should probably skim this section and consult it later if unfamiliar with some type or function.

```
module Maker where
```

We use a typical composition of monad transformers from the `mtl` library to structure stateful actions. This becomes relevant in section 4.2 (*The Maker monad*).

```
import Control.Monad.State (
  MonadState,    Type class of monads with state
  StateT,        Type constructor that adds state to a monad type
  execStateT,    Runs a state monad with given initial state
  get,           Gets the state in a do block
  put)           Sets the state in a do block

import Control.Monad.Reader (
  MonadReader,   Type class of monads with “environments”
  ask,           Reads the environment in a do block
  local)         Runs a sub-computation with a modified environment

import Control.Monad.Writer (
  MonadWriter,   Type class of monads that emit logs
  WriterT,       Type constructor that adds logging to a monad type
  runWriterT)    Runs a writer monad

import Control.Monad.Except (
  MonadError,    Type class of monads that fail
  Except,        Type constructor of failing monads
  throwError,    Short-circuits the monadic computation
  runExcept)     Runs a failing monad
```

Our numeric types use decimal fixed-point arithmetic.

```
import Data.Fixed (
    Fixed,           Type constructor for fixed-point numbers of given precision
    HasResolution (..)) Type class for specifying precisions
```

We rely on the `lens` library for accessing nested values. There is no need to understand the theory behind lenses to understand this program. The notation $a \circ b \circ c$ denotes a nested accessor much like `a.b.c` in C-style languages; for more details, consult lens documentation¹.

```
import Control.Lens (
    makeFields,      Defines lenses for record fields
    view, preview,    Reads a lens in a do block
    (&~),             Lets us use a do block with setters  $\diamond$  Get rid of this.
    ix,               Lens for map retrieval and updating
    at,               Lens for map insertion

    Operators for partial state updates in do blocks:
    (:=),             Replace
    (-=), (+=), (*=), Update arithmetically
    (%=),             Update according to function
    (?=))             Insert into map
```

Where the Solidity code uses `mapping`, we use Haskell’s regular tree-based map type².

```
import Data.Map (
    Map,           Type constructor for mappings
    ∅,             Polymorphic empty mapping
    singleton)     Creates a mapping with a single key–value pair
```

For sequences of log entries, we use a sequence structure which has better time complexity than regular lists.

```
import           Data.Sequence (Seq)
import qualified Data.Sequence as Sequence
```

Some less interesting imports are omitted from this document.

¹Gabriel Gonzalez’s 2013 article *Program imperatively using Haskell* is a good introduction.

²We assume the axiom that Keccak hash collisions are impossible.

Chapter 3

Types

3.1 Numeric types

Many Ethereum tokens (e.g. ETH, DAI, and MKR) are denominated with 18 decimals. That makes decimal fixed point with 18 digits of precision a natural choice for representing currency quantities. We call such quantities "wads" (as in "wad of cash").

For some quantities, such as the rate of deflation per second, we want as much precision as possible, so we use twice the number of decimals. We call such quantities "rays" (mnemonic "rate," but also imagine a very precisely aimed ray of light).

Dummy types for specifying precisions

data E18; **data** E36

Specify 10^{-18} as the precision of E18

instance HasResolution E18 **where**

resolution _ = $10 \uparrow (18 :: \text{Integer})$

Specify 10^{-36} as the precision of E36

instance HasResolution E36 **where**

resolution _ = $10 \uparrow (36 :: \text{Integer})$

Create the distinct wad type for currency quantities

newtype Wad = Wad (Fixed E18)

deriving (Ord, Eq, Num, Real, Fractional)

Create the distinct ray type for precise rate quantities

newtype Ray = Ray (Fixed E36)

deriving (Ord, Eq, Num, Real, Fractional)

In calculations that combine **wads** and **rays**, we have to convert between the number types. Haskell does not convert numbers automatically, so when we explicitly need it, we use a *cast* function.

Convert via fractional n/m form.
 $cast :: (Real\ a, Fractional\ b) \Rightarrow a \rightarrow b$
 $cast = fromRational \circ toRational$

We also define a type for non-negative integers.

```
newtype Nat = Nat Int
deriving (Eq, Ord, Enum, Num, Real, Integral)
```

3.2 Identifier type

There are several kinds of identifiers used in the system, and we can use types to distinguish them.

The type parameter a creates distinct types.
 For example, `Id Foo` and `Id Bar` are incompatible.

```
data Id  $a$  = Id String
deriving (Show, Eq, Ord)
```

We will often use mappings from IDs to the value type corresponding to that ID type, so we define an alias for such mappings.

```
type IdMap  $a$  = Map (Id  $a$ )  $a$ 
```

3.3 Domain types

This section introduces the records stored by the Maker system. The order of presentation is by use; types further down refer to types further up, but not the other way around.

```
data Address = Address String
deriving (Ord, Eq, Show)
```

We also have three predefined entities:

The DAI token address
 $id_{\text{DAI}} = \text{Id "Dai"}$

The CDP engine address
 $id_{\text{vat}} = \text{Address "Vat"}$

The account with ultimate authority
 \diamond *Kludge until authority is modelled*
 $id_{\text{god}} = \text{Address "God"}$

```
data Gem =
  Gem {
    gemTotalSupply :: !Wad,
    gemBalanceOf   :: !(Map Address Wad),
    gemAllowance   :: !(Map (Address, Address) Wad)
  } deriving (Eq, Show)
makeFields '' Gem
```

```
data Jar = Jar {
  Collateral token
  jarGem :: !Gem,

  Market price
  jarTag :: !Wad,

  Price expiration
  jarZzz :: !Nat
} deriving (Eq, Show)
makeFields '' Jar
```

```
data Ilk = Ilk {
  Collateral vault
  ilkJar :: !(Id Jar),

  Liquidation penalty
  ilkAxe :: !Ray,

  Debt ceiling
  ilkHat :: !Wad,
```

```

Liquidation ratio
  ilkMat :: !Ray,
Stability fee
  ilkTax :: !Ray,
Limbo duration
  ilkLag :: !Nat,
Last dripped
  ilkRho :: !Nat,
Total debt in dai
  ilkDin :: !Wad,
Price of debt coin
  ilkChi :: !Ray
} deriving (Eq, Show)
makeFields '' Ilk

```

```

data Urn = Urn {
  Address of biting cat
    urnCat :: !(Maybe Address),
  Address of liquidating vow
    urnVow :: !(Maybe Address),
  Issuer
    urnLad :: !Address,
  CDP type
    urnIlk :: !(Id Ilk),
  Outstanding debt in debt coins
    urnArt :: !Wad,
  Collateral amount in debt coins
    urnJam :: !Wad
} deriving (Eq, Show)
makeFields '' Urn

```

```

data Vat = Vat {
  Market price
    vatFix :: !Wad,

```



```

Sensitivity
  vatHow :: !Ray,

Target price
  vatPar  :: !Wad,

Target rate
  vatWay  :: !Ray,

Last prodded
  vatTau  :: !Nat,

Unprocessed revenue from stability fees
  vatPie  :: !Wad,

Bad debt from liquidated CDPs
  vatSin  :: !Wad,

Collateral tokens
  vatJars :: !(IdMap Jar),

CDP types
  vatIlks :: !(IdMap Ilk),

CDPs
  vatUrns :: !(IdMap Urn)
} deriving (Eq, Show)
makeFields '' Vat

```

```

data System =
  System {
    systemVat    :: Vat,
    systemEra    :: !Nat,
    systemSender :: Address
  } deriving (Eq, Show)
makeFields '' System

```

3.4 Default data

```

defaultIlk :: Id Jar → Ilk
defaultIlk idjar = Ilk {
  ilkJar = idjar,

```

```

    ilkAxe = Ray 1,
    ilkMat = Ray 1,
    ilkTax = Ray 1,
    ilkHat = Wad 0,
    ilkLag = Nat 0,
    ilkChi = Ray 1,
    ilkDin = Wad 0,
    ilkRho = Nat 0
  }

```

```

defaultUrn :: Id Ilk → Address → Urn
defaultUrn idilk idlad = Urn {
  urnVow = Nothing,
  urnCat = Nothing,
  urnLad = idlad,
  urnIlk = idilk,
  urnArt = Wad 0,
  urnJam = Wad 0
}

```

```

initialVat :: Ray → Vat
initialVat how0 = Vat {
  vatTau = 0,
  vatFix = Wad 1,
  vatPar = Wad 1,
  vatHow = how0,
  vatWay = Ray 1,
  vatPie = Wad 0,
  vatSin = Wad 0,
  vatIlks = ∅,
  vatUrns = ∅,
  vatJars =
    singleton idDAI Jar {
      jarGem = Gem {
        gemTotalSupply = 0,
        gemBalanceOf = ∅,
        gemAllowance = ∅
      },
      jarTag = Wad 0,
      jarZzz = 0
    }
}

```

```

    }
  }

```

```

initialSystem :: Ray → System
initialSystem how0 = System {
  systemVat      = initialVat how0,
  systemEra      = 0,
  systemSender   = idgod
}

```


Chapter 4

Act framework

4.1 Act descriptions

We define the Maker act vocabulary as a data type. This is used for logging and generally for representing acts.

```
data Act =  
  Bite (Id Urn)  
| Draw (Id Urn) Wad  
| Form (Id Ilk) (Id Jar)  
| Free (Id Urn) Wad  
| Frob Ray  
| Give (Id Urn) Address  
| Grab (Id Urn)  
| Heal Wad  
| Lock (Id Urn) Wad  
| Loot Wad  
| Mark (Id Jar) Wad      Nat  
| Open (Id Urn) (Id Ilk)  
| Prod  
| Poke (Id Urn)  
| Pull (Id Jar) Address Wad  
| Shut (Id Urn)  
| Tell Wad  
| Warp Nat  
| Wipe (Id Urn) Wad  
deriving (Eq, Show)
```

Acts which are logged through the **note** modifier record the sender ID and the act descriptor.

```
data Log = LogNote Address Act
deriving (Show, Eq)
```

Acts can fail. We divide the failure modes into general assertion failures and authentication failures.

```
data Error = AssertError | AuthError
deriving (Show, Eq)
```

4.2 The Maker monad

The reader does not need any abstract understanding of monads to understand the code. What they give us is a nice syntax—the **do** notation—for expressing exceptions, state, and logging in a way that is still purely functional.

```
newtype Maker a =
  Maker (StateT System
        (WriterT (Seq Log)
          (Except Error)) a)
deriving (
  Functor, Applicative, Monad,
  MonadError Error,
  MonadState System,
  MonadWriter (Seq Log)
)

exec :: System
     → Maker ()
     → Either Error (System, Seq Log)
exec sys (Maker m) =
  runExcept (runWriterT (execStateT m sys))

instance MonadReader System Maker where
  ask = Maker get
  local f (Maker m) = Maker $ do
    s ← get; put (f s)
    x ← m; put s
  return x
```

4.3 Constraints

```

type Reads r m = MonadReader r m
type Writes w m = MonadState w m
type Logs    m = MonadWriter (Seq Log) m
type Fails    m = MonadError Error m
type IsAct = ?act :: Act
type Notes    m = (IsAct, Logs m)

```

4.4 Accessor aliases

```

ilkAt id = vat ◦ ilks ◦ ix id
urnAt id = vat ◦ urns ◦ ix id
jarAt id = vat ◦ jars ◦ ix id

```

4.5 Logging and asserting

```

log :: Logs m ⇒ Log → m ()
log x = Writer.tell (Sequence.singleton x)
aver :: Fails m ⇒ Bool → m ()
aver x = unless x (throwError AssertionError)
need :: (Fails m, Reads r m)
  ⇒ Getting (First a) r a → m a
need f = preview f ≫ λcase
  Nothing → throwError AssertionError
  Just x → return x

```

4.6 Modifiers

```

note ::
  (IsAct, Logs m,
   Reads r m,

```

HasSender r Address)
 $\Rightarrow m\ a \rightarrow m\ a$

note $k = \mathbf{do}$
 $s \leftarrow \text{view sender}$
 $x \leftarrow k$
 $\text{log } (\text{LogNote } s\ ?act)$
 $\text{return } x$

auth ::
 (IsAct, Fails m ,
 Reads $r\ m$,
 HasSender r Address)
 $\Rightarrow m\ a \rightarrow m\ a$

auth continue = **do**
 $s \leftarrow \text{view sender}$
 $\text{unless } (s \equiv id_{god})$
 $(\text{throwError } \text{AuthError})$
 continue

Chapter 5

Acts

We call the basic operations of the Dai credit system "acts."

Table 5.1: Urn acts in the five stages of risk

	give	shut	lock	wipe	free	draw	bite	grab	plop	
Pride	•	•	•	•	•	•				overcollateralized
Anger	•	•	•	•	•					debt ceiling reached
Worry	•	•	•	•						price feed in limbo
Panic	•	•	•	•			•			undercollateralized
Grief	•							•		liquidation initiated
Dread	•								•	liquidation in progress

5.1 Risk assessment

We divide an urn’s situation into five stages of risk. Table 5.1 shows which acts each stage allows. The stages are naturally ordered from more to less risky.

```
data Stage = Dread | Grief | Panic | Worry | Anger | Pride
deriving (Eq, Ord, Show)
```

First we define a pure function *analyze* that determines an urn’s stage.

```
analyze era0 par0 urn0 ilk0 jar0 =
  let
    cap = view din ilk0 * cast (view chi ilk0)
    pro = view jam urn0 * view tag jar0
    con = view art urn0 * cast (view chi ilk0) * par0
    min = con * view mat ilk0
  in if
    Undergoing liquidation?
      | view vow urn0 ≠ Nothing → Dread
    Liquidation triggered?
      | view cat urn0 ≠ Nothing → Grief
    Undercollateralized?
      | pro < min → Panic
    Price feed expired?
      | era0 > view zzz jar0 + view lag ilk0 → Panic
    Price feed in limbo?
      | view zzz jar0 < era0 → Worry
    Debt ceiling reached?
      | cap > view hat ilk0 → Anger
    Safely overcollateralized.
      | otherwise → Pride
```

Now we define the internal act **gaze** which returns the value of *analyze* after ensuring the system state is updated.

```

gaze  $id_{\text{urn}}$  = do
  prod
   $id_{\text{ilk}} \leftarrow \text{need } (\text{urnAt } id_{\text{urn}} \circ \text{ilk})$ 
  drip  $id_{\text{ilk}}$ 
   $\text{era}_0 \leftarrow \text{view era}$ 
   $\text{par}_0 \leftarrow \text{view } (\text{vat} \circ \text{par})$ 
   $\text{urn}_0 \leftarrow \text{need } (\text{urnAt } id_{\text{urn}})$ 
   $\text{ilk}_0 \leftarrow \text{need } (\text{ilkAt } (\text{view ilk urn}_0))$ 
   $\text{jar}_0 \leftarrow \text{need } (\text{jarAt } (\text{view jar ilk}_0))$ 
  return (analyze  $\text{era}_0$   $\text{par}_0$   $\text{urn}_0$   $\text{ilk}_0$   $\text{jar}_0$ )

```

5.2 Lending

```

open  $id_{\text{urn}}$   $id_{\text{ilk}}$  =
  note $ do
     $id_{\text{lad}} \leftarrow \text{view sender}$ 
     $\text{vat} \circ \text{urns} \circ \text{at } id_{\text{urn}} \text{ ?= defaultUrn } id_{\text{ilk}} id_{\text{lad}}$ 

```

```

lock  $id_{\text{urn}}$   $x$  =
  note $ do
    Ensure CDP exists; identify collateral type
     $id_{\text{ilk}} \leftarrow \text{need } (\text{urnAt } id_{\text{urn}} \circ \text{ilk})$ 
     $id_{\text{jar}} \leftarrow \text{need } (\text{ilkAt } id_{\text{ilk}} \circ \text{jar})$ 
    Record an increase in collateral
     $\text{urnAt } id_{\text{urn}} \circ \text{jam} += x$ 
    Take sender's tokens
     $id_{\text{lad}} \leftarrow \text{view sender}$ 
    pull  $id_{\text{jar}}$   $id_{\text{lad}}$   $x$ 

```

```

free  $id_{\text{urn}}$   $\text{wad}_{\text{gem}}$  =

```

note \$ do

Fail if sender is not the CDP owner.

$id_{sender} \leftarrow view\ sender$

$id_{lad} \leftarrow need\ (urnAt\ id_{urn} \circ lad)$

aver $(id_{sender} \equiv id_{lad})$

Tentatively record the decreased collateral.

$urnAt\ id_{urn} \circ jam \text{ -- } wad_{gem}$

Fail if collateral decrease results in undercollateralization.

gaze $id_{urn} \gg= aver \circ (\equiv Pride)$

Send the collateral to the CDP owner.

$id_{ilk} \leftarrow need\ (urnAt\ id_{urn} \circ ilk)$

$id_{jar} \leftarrow need\ (ilkAt\ id_{ilk} \circ jar)$

push $id_{jar}\ id_{lad}\ wad_{gem}$

draw $id_{urn}\ wad_{DAI} =$

note \$ do

Fail if sender is not the CDP owner

$id_{sender} \leftarrow view\ sender$

$id_{lad} \leftarrow need\ (urnAt\ id_{urn} \circ lad)$

aver $(id_{sender} \equiv id_{lad})$

Update price of debt coin

$id_{ilk} \leftarrow need\ (urnAt\ id_{urn} \circ ilk)$

$chi_1 \leftarrow drip\ id_{ilk}$

Denominate draw amount in debt coin

let $wad_{chi} = wad_{DAI} / cast\ chi_1$

Increase debt

$urnAt\ id_{urn} \circ art \text{ += } wad_{chi}$

Roll back unless overcollateralized

gaze $id_{urn} \gg= aver \circ (\equiv Pride)$

Mint dai and send to the CDP owner

mint $id_{DAI}\ wad_{DAI}$

push $id_{DAI}\ id_{lad}\ wad_{DAI}$

wipe $id_{urn}\ wad_{DAI} =$

note \$ do

Fail if sender is not the CDP owner
 $id_{sender} \leftarrow view\ sender$
 $id_{lad} \leftarrow need\ (urnAt\ id_{urn} \circ lad)$
 $aver\ (id_{sender} \equiv id_{lad})$
 Update price of debt coin
 $id_{ilk} \leftarrow need\ (urnAt\ id_{urn} \circ ilk)$
 $chi_1 \leftarrow drip\ id_{ilk}$
 Denominate dai amount in debt coin
 $let\ wad_{chi} = wad_{DAI} / cast\ chi_1$
 Roll back if the CDP is not overcollateralized
 $gaze\ id_{urn} \gg= aver \circ (\equiv\ Pride)$
 Reduce debt
 $urnAt\ id_{urn} \circ art \ -=\ wad_{chi}$
 Take dai from CDP owner, or roll back
 $pull\ id_{DAI}\ id_{lad}\ wad_{DAI}$
 Destroy dai
 $burn\ id_{DAI}\ wad_{DAI}$

$give\ id_{urn}\ id_{lad} =$
 $note\ \$\ do$
 $x \leftarrow need\ (urnAt\ id_{urn} \circ lad)$
 $y \leftarrow view\ sender$
 $aver\ (x \equiv y)$
 $urnAt\ id_{urn} \circ lad := id_{lad}$

$shut\ id_{urn} =$
 $note\ \$\ do$
 Update price of debt coin
 $id_{ilk} \leftarrow need\ (urnAt\ id_{urn} \circ ilk)$
 $chi_1 \leftarrow drip\ id_{ilk}$
 Attempt to repay all the CDP's outstanding dai
 $art0 \leftarrow need\ (urnAt\ id_{urn} \circ art)$
 $wipe\ id_{urn}\ (art0 * cast\ chi_1)$
 Reclaim all the collateral
 $jam0 \leftarrow need\ (urnAt\ id_{urn} \circ jam)$
 $free\ id_{urn}\ jam0$

Nullify the CDP

$\text{vat} \circ \text{urns} \circ \text{at } id_{\text{urn}} := \text{Nothing}$

5.3 Frequent adjustments

```

prod = note $ do
  era0 ← view era
  tau0 ← view (vat ∘ tau)
  fix0 ← view (vat ∘ fix)
  par0 ← view (vat ∘ par)
  how0 ← view (vat ∘ how)
  way0 ← view (vat ∘ way)
  let
    Time difference in seconds
    fan = era0 - tau0
    Current deflation rate applied to target price
    par1 = par0 * cast (way0 ↑↑ fan)
    Sensitivity parameter applied over time
    wag = how0 * fromIntegral fan
    Deflation rate scaled up or down
    way1 = inj (prj way0 +
                  if fix0 < par0 then wag else - wag)
  vat ∘ par := par1
  vat ∘ way := way1
  vat ∘ tau := era0
  where
    Convert between multiplicative and additive form
    prj x = if x ≥ 1 then x - 1 else 1 - 1 / x
    inj x = if x ≥ 0 then x + 1 else 1 / (1 - x)

```

This internal act happens on every *poke*. It is also invoked when governance changes the **tax** of an **ilk**.

```

drip idilk = do
  Current time stamp
  era0 ← view era
  rho0 ← need (ilkAt idilk ∘ rho)
  Current stability fee
  tax0 ← need (ilkAt idilk ∘ tax)
  Current price of debt coin

```

```

chi0 ← need (ilkAt idilk ∘ chi)
let
  age = era0 − rho0
  chi1 = chi0 * tax0 ↑↑ age
  ilkAt idilk ∘ chi := chi1
  ilkAt idilk ∘ rho := era0
return chi1

```

5.4 Governance

```

form idilk idjar =
  auth ∘ note $ do
    vat ∘ ilks ∘ at idilk ?= defaultIlk idjar

```

```

frob how' =
  auth ∘ note $ do
    vat ∘ how := how'

```

5.5 Price feedback

```

mark idjar tag1 zzz1 =
  auth ∘ note $ do
    jarAt idjar ∘ tag := tag1
    jarAt idjar ∘ zzz := zzz1

```

```

tell x =
  auth ∘ note $ do
    vat ∘ fix := x

```


5.6 Liquidation and settlement

```

bite  $id_{\text{urn}}$  =
  note $ do
    Fail if urn is not undercollateralized
    gaze  $id_{\text{urn}}$   $\gg$  aver  $\circ$  ( $\equiv$  Panic)
    Record the sender as the requester of liquidation
     $id_{\text{cat}} \leftarrow \text{view sender}$ 
     $\text{urnAt } id_{\text{urn}} \circ \text{cat} := id_{\text{cat}}$ 
    Read current debt
     $\text{art0} \leftarrow \text{need } (\text{urnAt } id_{\text{urn}} \circ \text{art})$ 
    Update price of debt coin
     $id_{\text{ilk}} \leftarrow \text{need } (\text{urnAt } id_{\text{urn}} \circ \text{ilk})$ 
     $\text{chi}_1 \leftarrow \text{drip } id_{\text{ilk}}$ 
    Read liquidation penalty ratio
     $id_{\text{ilk}} \leftarrow \text{need } (\text{urnAt } id_{\text{urn}} \circ \text{ilk})$ 
     $\text{axe}_0 \leftarrow \text{need } (\text{ilkAt } id_{\text{ilk}} \circ \text{axe})$ 
    Apply liquidation penalty to debt
    let  $\text{art1} = \text{art0} * \text{axe}_0$ 
    Update debt and record it as in need of settlement
     $\text{urnAt } id_{\text{urn}} \circ \text{art} := \text{art1}$ 
    sin  $\quad \quad \quad \vdash = \text{art1} * \text{chi}_1$ 

```

```

grab  $id_{\text{urn}}$  =
  auth  $\circ$  note $ do
    Fail if CDP is not marked for liquidation
    gaze  $id_{\text{urn}}$   $\gg$  aver  $\circ$  ( $\equiv$  Grief)
    Record the sender as the CDP's settler
     $id_{\text{vow}} \leftarrow \text{view sender}$ 
     $\text{urnAt } id_{\text{urn}} \circ \text{vow} := id_{\text{vow}}$ 
    Clear the CDP's requester of liquidation
     $\text{urnAt } id_{\text{urn}} \circ \text{cat} := \text{Nothing}$ 

```

```

heal wadDAI =

```

```

auth ◦ note $ do
  vat ◦ sin == wadDAI

loot wadDAI =
  auth ◦ note $ do
    vat ◦ pie == wadDAI

```

5.7 Minting, burning, and transferring

```

pull idjar idlad w = do
  g ← need (jarAt idjar ◦ gem)
  g' ← transferFrom idlad idvat w g
  jarAt idjar ◦ gem := g'

push idjar idlad w = do
  g ← need (jarAt idjar ◦ gem)
  g' ← transferFrom idvat idlad w g
  jarAt idjar ◦ gem := g'

mint idjar wad0 = do
  jarAt idjar ◦ gem ◦ totalSupply += wad0
  jarAt idjar ◦ gem ◦ balanceOf ◦ ix idvat += wad0

burn idjar wad0 = do
  jarAt idjar ◦ gem ◦ totalSupply -= wad0
  jarAt idjar ◦ gem ◦ balanceOf ◦ ix idvat -= wad0

```

5.8 Test-related manipulation

```

warp t =
  auth ◦ note $ do
    era += t

```

5.9 Other stuff

```

perform :: Act → Maker ()
perform x =
  let ?act = x in case x of
    Form id jar   → form id jar
    Mark jar tag zzz → mark jar tag zzz
    Open id ilk   → open id ilk
    Tell wad      → tell wad
    Frob ray      → frob ray
    Prod          → prod
    Warp t        → warp t
    Give urn lad  → give urn lad
    Pull jar lad wad → pull jar lad wad
    Lock urn wad  → lock urn wad

transferFrom
  :: (MonadError Error m)
  ⇒ Address → Address → Wad
  → Gem → m Gem

transferFrom src dst wad gem =
  case view (balanceOf ∘ at src) gem of
    Nothing →
      throwError AssertionError
    Just balance → do
      aver (balance ≥ wad)
      return $ gem &~ do
        balanceOf ∘ ix src -= wad
        balanceOf ∘ at dst %=
          (λcase
            Nothing → Just wad
            Just x   → Just (wad + x))

```


Chapter 6

Testing

Appendix A

Act type signatures

```
type Numbers wad ray nat =  
  (wad~Wad, ray~Ray, nat~Nat)
```

We see that `drip` may fail; it reads an `ilk`'s `tax`, `cow`, `rho`, and `bag`; and it writes those same parameters except `tax`.

```
drip ::  
  (Fails m,  
   Reads r m,  
   HasEra r Nat,  
   HasVat r vatr,  
   HasIlks vatr (Map (Id Ilk) ilkr),  
   HasTax ilkr Ray,  
   HasRho ilkr Nat,  
   HasChi ilkr Ray,  
  Writes w m,  
  HasVat w vatw,  
  HasIlks vatw (Map (Id Ilk) ilkw),  
  HasRho ilkw Nat,  
  HasChi ilkw Ray)  
⇒ Id Ilk → m Ray
```

```
form ::  
  (IsAct, Fails m, Logs m,  
   Reads r m, HasSender r Address,  
   Writes w m, HasVat w vatw,
```

$\text{HasIlks } \text{vat}_w (\text{IdMap } \text{Ilk}))$
 $\Rightarrow \text{Id } \text{Ilk} \rightarrow \text{Id } \text{Jar} \rightarrow m \ ()$

$\text{frob} :: (\text{IsAct}, \text{Fails } m, \text{Logs } m,$
 $\quad \text{Reads } r \ m, \text{ HasSender } r \ \text{Address},$
 $\quad \text{Writes } w \ m, \text{ HasVat } w \ \text{vat}_w,$
 $\quad \text{HasHow } \text{vat}_w \ \text{ray})$
 $\Rightarrow \text{ray} \rightarrow m \ ()$

$\text{open} ::$
 $(\text{IsAct}, \text{Logs } m,$
 $\quad \text{Reads } r \ m, \text{ HasSender } r \ \text{Address},$
 $\quad \text{Writes } w \ m, \text{ HasVat } w \ \text{vat}_w,$
 $\quad \text{HasUrns } \text{vat}_w (\text{IdMap } \text{Urn}))$
 $\Rightarrow \text{Id } \text{Urn} \rightarrow \text{Id } \text{Ilk} \rightarrow m \ ()$

$\text{give} ::$
 $(\text{IsAct}, \text{Fails } m, \text{Logs } m,$
 $\quad \text{Reads } r \ m, \text{ HasSender } r \ \text{Address},$
 $\quad \text{HasVat } r \ \text{vat}_r,$
 $\quad \text{HasUrns } \text{vat}_r (\text{Map } (\text{Id } \text{Urn}) \ \text{urn}_r),$
 $\quad \text{HasLad } \text{urn}_r \ \text{Address},$
 $\quad \text{Writes } w \ m, \text{ HasVat } w \ \text{vat}_r)$
 $\Rightarrow \text{Id } \text{Urn} \rightarrow \text{Address} \rightarrow m \ ()$

$\text{lock} ::$
 $(\text{IsAct}, \text{Fails } m, \text{Logs } m,$
 $\quad \text{Reads } r \ m,$
 $\quad \text{HasSender } r \ \text{Address},$
 $\quad \text{HasVat } r \ \text{vat}_r,$
 $\quad \text{HasUrns } \text{vat}_r (\text{Map } (\text{Id } \text{Urn}) \ \text{urn}_r),$
 $\quad \text{HasIlk } \text{urn}_r (\text{Id } \text{Ilk}),$
 $\quad \text{HasIlks } \text{vat}_r (\text{Map } (\text{Id } \text{Ilk}) \ \text{ilk}_r),$
 $\quad \text{HasJar } \text{ilk}_r (\text{Id } \text{Jar}),$
 $\quad \text{HasJars } \text{vat}_r (\text{Map } (\text{Id } \text{Jar}) \ \text{jar}_r),$
 $\quad \text{HasGem } \text{jar}_r \ \text{Gem},$
 $\quad \text{Writes } w \ m,$
 $\quad \text{HasVat } w \ \text{vat}_w,$


```

    HasJars vatw (Map (Id Jar) jarr),
    HasUrns vatw (Map (Id Urn) urnw),
    HasJam urnw Wad)
⇒ Id Urn → Wad → m ()

```

```

mark ::
  (IsAct, Fails m, Logs m,
   Reads r m, HasSender r Address,
   Writes w m, HasVat w vatw,
    HasJars vatw (Map (Id Jar) jarw),
    HasTag jarw wad,
    HasZzz jarw nat)
⇒ Id Jar → wad → nat → m ()

```

```

tell ::
  (IsAct, Fails m, Logs m,
   Reads r m, HasSender r Address,
   Writes w m, HasVat w vatw,
    HasFix vatw wad)
⇒ wad → m ()

```

```

prod ::
  (IsAct, Logs m,
   Reads r m,
    HasSender r Address,
    HasEra r nat,
    HasVat r vatr, (HasPar vatr wad,
                      HasTau vatr nat,
                      HasHow vatr ray,
                      HasWay vatr ray,
                      HasFix vatr wad),
   Writes w m,
    HasVat w vatw, (HasPar vatw wad,
                      HasWay vatw ray,
                      HasTau vatw nat),
   Integral nat,
   Ord wad, Fractional wad,
   Fractional ray, Real ray)
⇒ m ()

```

```

warp ::
  (IsAct, Fails  $m$ , Logs  $m$ ,
   Reads  $r$   $m$ , HasSender  $r$  Address,
   Writes  $w$   $m$ , HasEra  $w$  nat,
   Num nat)
   $\Rightarrow$  nat  $\rightarrow m$  ()

pull ::
  (Fails  $m$ ,
   Reads  $r$   $m$ ,
   HasVat  $r$  vatr, HasJars vatr (Map (Id Jar) jarr),
   HasGem jarr Gem,
   Writes  $w$   $m$ ,
   HasVat  $w$  vatw, HasJars vatw (Map (Id Jar) jarr))
   $\Rightarrow$  Id Jar  $\rightarrow$  Address  $\rightarrow$  Wad  $\rightarrow m$  ()

push ::
  (Fails  $m$ ,
   Reads  $r$   $m$ ,
   HasVat  $r$  vatr, HasJars vatr (Map (Id Jar) jarr),
   HasGem jarr Gem,
   Writes  $w$   $m$ ,
   HasVat  $w$  vatw, HasJars vatw (Map (Id Jar) jarr))
   $\Rightarrow$  Id Jar  $\rightarrow$  Address  $\rightarrow$  Wad  $\rightarrow m$  ()

mint ::
  (Fails  $m$ ,
   Writes  $w$   $m$ ,
   HasVat  $w$  vatw, HasJars vatw (Map (Id Jar) jarr),
   HasGem jarr gemr,
   HasTotalSupply gemr Wad,
   HasBalanceOf gemr (Map Address Wad))
   $\Rightarrow$  Id Jar  $\rightarrow$  Wad  $\rightarrow m$  ()

burn ::
  (Fails  $m$ ,
   Writes  $w$   $m$ ,
   HasVat  $w$  vatw, HasJars vatw (Map (Id Jar) jarr),
   HasGem jarr gemr,

```

HasTotalSupply *gem_r* Wad,
 HasBalanceOf *gem_r* (Map Address Wad))
 $\Rightarrow \text{Id Jar} \rightarrow \text{Wad} \rightarrow m ()$

grab ::

(IsAct, Fails *m*, Logs *m*,
 Numbers wad ray nat,
 Reads *r m*,
 HasSender *r* Address,
 HasEra *r* Nat,
 HasVat *r vat_r*,
 HasFix vat_r wad,
 HasPar vat_r wad,
 HasHow vat_r ray,
 HasWay vat_r ray,
 HasTau vat_r nat,
 HasUrns vat_r (Map (Id Urn) urn_r),
 HasJam urn_r wad,
 HasArt urn_r wad,
 HasCat urn_r (Maybe Address), HasVow urn_r (Maybe Address),
 HasIlk urn_r (Id Ilk),
 HasIlks vat_r (Map (Id Ilk) ilk_r),
 HasHat ilk_r wad,
 HasMat ilk_r wad,
 HasDin ilk_r wad,
 HasTax ilk_r ray,
 HasLag ilk_r nat,
 HasChi ilk_r ray, HasRho ilk_r nat,
 HasJar ilk_r (Id Jar),
 HasJars vat_r (Map (Id Jar) jar_r),
 HasGem jar_r Gem,
 HasTag jar_r wad,
 HasZzz jar_r nat,
 Writes *w m*,
 HasVat *w vat_w*,
 HasTau vat_w nat,
 HasWay vat_w ray, HasPar vat_w wad,
 HasUrns vat_w (Map (Id Urn) urn_w),
 HasJam urn_w wad, HasArt urn_w wad,
 HasVow urn_w Address,
 HasCat urn_w (Maybe Address),

```

HasIlks vatw (Map (Id Ilk) ilkw),
  HasChi ilkw ray,
  HasRho ilkw nat,
  HasJars vatw (Map (Id Jar) jarr)
) ⇒ Id Urn → m ()

```